
ABSTRACT
The DATA step can write information to virtually any destination. You are probably familiar with writing to SAS data
sets and external files. But email can also function as a destination.
This paper will discuss how to configure the email features in the system options. Then we will proceed with sending a
simple email from the data step, with or without attachments. Next we will use extensions to the standard PUT
statement that support the email facility to send personalized mass-mailings. Finally we will show how to send
procedure output that has been created using ODS.

INTRODUCTION
The DATA Step is the core of the data processing facilities in SAS. It is used to read information from virtually any
source. It is used to process the data any way you like and eventually to write the processed data to virtually any
destination.
One of those destinations may be an email server, meaning that you can send emails directly from the DATA Step. In
order to do that, you first have to define the email environment that SAS will use. This environment is defined in a
number of system options.
Once the environment is set up, sending email is like writing information to an external destination, using FILENAME,
FILE and PUT statements.

SETTING UP TO SEND E-MAIL
SAS supports three interface methods to send mail:
•
MAPI: "Messaging Application Program Interface". This is the default and makes use of the mail client on your
computer, e.g. Microsoft Exchange or Outlook or Mozilla Thunderbird.
•
SMTP: "Simple Mail Transfer Protocol". With SMTP you bypass the mail client and you have to specify outgoing
mail account settings, comparable to specifying the outgoing mail settings in Outlook or Thunderbird.
•
VIM: "Vendor Independent Mail". VIM can be used in combination with Lotus Notes or cc:Mail.
You have to specify the options to define the mail interface in the SAS configuration file (for SAS 9: SASV9.CFG),
since they are used in the initialization of the SAS session. This file is normally located in
C:\Program Files\SAS\SAS 9.1\nls\en. Look in the System Options in the Display Manager (Tools Æ Options Æ
System Options Æ System Administration Æ Installation) to verify the location in your installation.
To find out your current email settings, you also go to System options Æ Communications Æ email (Figure 1)
Figure 1: The SAS Email options

SETTING UP THE MAPI INTERFACE

MAPI is the default method to be used by SAS. The MAPI interface works with the mail system as installed on the
computer where SAS runs. It retrieves the necessary information from that system, so you don't need to specify
anything. Just verify that the system option EMAILSYS has the value MAPI and you are ready to go.
SETTING UP THE SMTP INTERFACE

SMTP bypasses the mail system that is present on the computer and sends mails directly. So you have to specify
information similar to the information you specify for your outgoing mail account in mail clients like Outlook or
Thunderbird. So look under "preferences" or "outgoing mail settings" or a similar heading in your mail client to find
the right settings. The options you have to set are:
-EMAILSYS
-EMAILHOST
-EMAILID
-EMAILPW

SMTP
name of outgoing mail server
"sender information"
"password to access mail server"

The sender information and password have to be enclosed in double quotes if they contain blanks.
For example:
-EMAILSYS SMTP
-EMAILHOST mail.planet.nl
-EMAILID "Erik Tilanus <erik.tilanus@planet.nl>"
-EMAILPW etila999
There are two more options that can be set: EMAILPORT and EMAILAUTHPROTOCOL. EMAILPORT defines the
port to which the SMTP server listens. The default port is 25, which will be correct in most cases.
EMAILAUTHPROTOCOL defines whether you need a login sequence for the outgoing mail server.
SETTING UP THE VIM INTERFACE

For the VIM interface to work, you have to specify the EMAILSYS, EMAILID and EMAILPW options:
-EMAILSYS
VIM
-EMAILID
User log on code
-EMAILPW
User password
By the way, if you are sending mails from the Display Manager menu (File Æ Send Mail…) SAS will use either the
MAPI or the VIM interface and not the SMTP interface.

SENDING A MAIL MESSAGE FROM A DATA STEP
Once you have set up the system options, you use a FILENAME statement to define the output destination, a FILE
statement to link your PUT statements to the right output destination and PUT statements to create the message. :
Program 1: Basic DATA step to send mail.
FILENAME Mailbox EMAIL 'Erik.Tilanus@planet.nl'
Subject='Test Mail message';
DATA _NULL_;
FILE Mailbox;
PUT "Hello";
PUT "This is a message from the DATA step";
RUN;
You don't need to specify the addressee and the subject in the FILENAME statement; you can do it also in the FILE
statement:
Program 2: As Program 1, but with all mail data moved from FILENAME to FILE statement
FILENAME Mailbox EMAIL;
DATA _NULL_;
FILE Mailbox TO='Erik_Tilanus@compuserve.com'
SUBJECT='Test Mail message';
PUT "Hello";
PUT "This is a message from the DATA step";
RUN;
In the same way you can enter also copyreaders, by specifying CC=mailaddress or BCC=mailaddress.
If you want to send a mail to more than one address, you list all addresses (each between quotes) within

parentheses.
You can also include a display name next tot the mail address. In that case the mail address is placed between <…>
and the display name precedes it.
If you are using the SMTP interface, you can overrule the default sender and specify a separate reply address. Use
the FROM option and the REPLYTO option to accomplish this.
All these options are put together in the following example:
FILE MailBox TO=('Erik <Erik_Tilanus@compuserve.com>'
'Myself <Erik.Tilanus@planet.nl>')
CC=('Synchrona@planet.nl'
'Erik_Tilanus@hotmail.com')
BCC='Anneke.Tilanus@planet.nl'
FROM='My Business <Synchrona@planet.nl>'
REPLYTO='My anonymous address <villavinkeveen@hotmail.com>'
SUBJECT='Test Mail message to group and other options';
Messages are sent upon completion of the DATA step, not per observation. But off course the PUT statements will be
executed with each iteration of the DATA step and therefore continue to build the message.
If your outgoing mail server is not picky, you can send anonymous mails this way. To demonstrate this I set the
EMAILID to "unknown@nowhere.com". Then I used Program 2 and the extended FILE statement to send a mail.
Figure 2 shows the mail as received by me. I also checked what happened if I leave EMAILID blank. Then SAS will
use the Windows user and system name as the sender information.
Figure 2: Mail received from Program 2 and the extended FILE statement

If you want to build more than one message simultaneously for different addressees, you may specify more
FILENAME statements and direct the PUT statement to either of them with different FILE statements. All messages
will be sent at the end of the data step. Program 3 shows the principle.

Program 3: Sending more mails simultaneously
FILENAME MailBox1 EMAIL;
FILENAME MailBox2 EMAIL;
FILENAME MailBox3 EMAIL;
DATA _NULL_;
FILE MailBox1 TO='Erik <Erik_Tilanus@compuserve.com>'
SUBJECT='First addressee';
PUT "Hello Mailbox1";
FILE MailBox2 TO='erik.tilanus@planet.nl'
SUBJECT='Second addressee';
PUT "Hello Mailbox2 ";
FILE MailBox3 TO='synchrona@planet.nl'
SUBJECT='Third addressee';
PUT "Hello Mailbox3 ";
RUN;

SENDING ATTACHMENTS
Including attachments is also rather straightforward: specify the attachment in either the FILENAME or FILE
statement with the keyword ATTACH, like:
FILENAME MailBox EMAIL ATTACH='C:\SASUtil\Tips.doc';
or
FILE Mailbox ATTACH='C:\SASUtil\Tips.doc';
If you want to specify the attachment under program control, you can use the special option in the PUT statement:
PUT '!EM_ATTACH!' AttachInfo;
where AttachInfo is either a character variable that contains the full path and name of the file that should be attached
or the full path and name between quotes. More about this special PUT directive and other similar constructions are
discussed below in the paragraph "Personalized bulk mail".

SENDING OUTPUT
Output can be mailed in two ways. A simple and straightforward method is to write the output to a file and then specify
that file as an attachment in a subsequent DATA _NULL_ step that sends the mail. If you don't want to specify the full
file name in this second step, you can derive it from the fileref as is done in Program 4. The first DATA step in this
program produces the report. The enclosing ODS statements specify that the report is to be created in RTF format.
The second DATA step picks up the file name, by opening the file with the FOPEN function and getting the full name
using the FINFO function. Thus it assigns the full file name to the variable Attachment and attaches the file to the
mail.
It is obvious that the second step is a nice candidate to be wrapped in a macro.
To send procedure output you can off course send the output to a file using ODS and then attach the file in the same
way as in Program 4.
If you don't need the file on your computer once it has been mailed, you could use a TEMP file, however SAS
attaches the suffix .txt to TEMP files. The receiver has to change the suffix into .rtf to open it properly in MS Word or
similar.
Program 5 shows another method: you can use the mail destination directly in the ODS statement.
There is however a distinct difference between the approach of Program 4 and Program 5. In Program 4 the output is
sent as an attachment, while in Program 5 the output is sent in-line as the body of the message.
The latter works fine with the HTML and RTF destinations, but not with most other destinations. HTML output will be
displayed in the mail message. RTF output however would display as a text message showing all the RTF
commands. To display the output correctly, you will have to save the contents of the message as a text file with
extension RTF and then reopen it as RTF document with e.g. MS Word.

Program 4: Creating a report in a DATA step and sending it as attachment
FILENAME MailBox EMAIL 'Erik.Tilanus@planet.nl'
SUBJECT='Mail message with RTF attachment';
FILENAME rtffile 'e:\testlist.rtf';
ODS LISTING CLOSE;
ODS RTF FILE=rtffile;
DATA _NULL_;
FILE PRINT;
PUT "This is my report";
RUN;
ODS RTF CLOSE;
ODS LISTING;
DATA _NULL_;
attachment=FINFO(FOPEN('rtffile'),'File Name');
FILE MailBox;
PUT "attached you find the report";
PUT '!EM_ATTACH!' attachment;
RUN;
Program 5: Sending procedure output directly by mail (HTML or RTF)
FILENAME mail EMAIL TO="erik.tilanus@planet.nl"
SUBJECT="HTML OUTPUT" CONTENT_TYPE="text/html";
ODS LISTING CLOSE;
ODS HTML BODY=mail;
PROC PRINT DATA=mydata.Sales;
RUN;
ODS HTML CLOSE;
ODS LISTING;
Figure 3: The mail as sent by Program 5

PERSONALIZED BULK MAIL
In the previous paragraphs one message was sent per output destination (FILENAME statement), upon completion of
the DATA step. If you want to send a mail per observation or per BY-group or similar, you need a PUT statement with
a special directive:
PUT '!EM_directive!' specification;
Table 1 shows the possible directives. The included examples in this table show quoted text as directive and
specification. However that is not required. You could as well use character variables containing the appropriate
information. It's a PUT statement, remember!
You can have only one directive per line and it should be the first item, otherwise it is not recognized. So the obvious
solution is to use separate PUT statements per directive, but you can also use the "/" within the PUT statement to
move to the next line.
The !EM_NEWMSG! directive clears all previous information and lets you start a new message. If you leave it out,
every new message will build upon the previous one, just adding the information rather than replacing it.
The !EM_ABORT! prevents sending a message at termination of the DATA step. If you leave it out, but you have
!EM_NEWMSG! in, an empty message will be sent to the address in the FILENAME or FILE statement.
Table 1: The special PUT statements to manage sending of mail messages.

Directive
!EM_TO!
!EM_CC!
!EM_BCC!
!EM_SUBJECT!
!EM_ATTACH!
!EM_SEND!
!EM_NEWMSG!
!EM_ABORT!

Specification
Mail address, as in
FILENAME TO=
Mail address, as in
FILENAME CC=
Mail address, as in
FILENAME BCC=
Subject for message
between quotes
Full file address, as in
FILENAME ATTACH=
None
None
None

Example
PUT '!EM_TO!' "erik.tilanus@planet.nl";
PUT '!EM_CC!' ("erik.tilanus@planet.nl"
"erik_tilanus@hotmail.com");
PUT '!EM_BCC!' "synchrona@planet.nl";
PUT '!EM_SUBJECT!' "My SAS tips"";
PUT '!EM_ATTACH!' "C:\SASUtil\Tips.doc";
PUT '!EM_SEND!';
PUT '!EM_NEWMSG!';
PUT '!EM_ABORT!';

EXAMPLE BULK MAIL

Assume you have a travel agency chain and you want to distribute the latest price information selectively to all branch
offices. Each branch office receives the price information for packages at their location. The data set Forum.OfficeList
contains the names of the offices, their destination code and their email addresses. For some destinations there are
also special packages, for internet sales only. The price information is contained in another data set (Forum.Fares).
This data set contains the destination airport, the level of the offered package (Luxury, Economy, Budget) and the
applicable price. Program 6 shows how this mailing could be accomplished. We read the OfficeList data set and for
each observation we set up an email. Then we read the Fares data set in a DO loop to select the right observations
and add them to the mail.
Listing 1 shows a part of the SAS log with the information about the sent mails. Note that I have used several of my
own mail addresses to deliver the mails to.

Program 6: Sending bulk mail
FILENAME mail EMAIL;
DATA _NULL_;
SET forum.Officelist END=eof ;
FILE mail;
PUT '!EM_TO!' mailaddress;
PUT '!EM_SUBJECT!' 'New price information for ' office;
PUT "These are the fares as valid on &SYSDATE";
PUT 'Selection: ' location;
DO n=1 to NOBS;
SET forum.fares POINT=n NOBS=nobs;
IF (location = 'XXX' and package = 'Special') OR
(location ne 'XXX' and package ne 'Special' and
location eq destination)
THEN PUT destination package price;
END;
PUT '!EM_SEND!' / '!EM_NEWMSG!';
IF eof THEN PUT '!EM_ABORT!';
RUN;
Listing 1: Partial SAS log of a bulk mailing
NOTE: The file MAIL is:
E-Mail Access Device
Message sent
To:
athens@travelpartners.eu
Cc:
Bcc:
Subject:
New price information for Athens
Attachments:
Message sent
To:
frankfurt@travelpartners.eu
Cc:
Bcc:
Subject:
New price information for Frankfurt
Attachments:
Message sent
To:
paris@travelpartners.eu
Cc:
Bcc:
Subject:
New price information for Paris
Attachments:
...
Message sent
To:
internet@travelpartners.eu
Cc:
Bcc:
Subject:
New price information for Internet
Attachments:
Message sent
To:
newyork@travelpartners.eu
Cc:
Bcc:
Subject:
New price information for NewYork
Attachments:
NOTE: 83 records were written to the file MAIL.
The minimum record length was 9.
The maximum record length was 47.
NOTE: There were 6 observations read from the data set FORUM.OFFICELIST.
NOTE: DATA statement used (Total process time):
real time
0.99 seconds
cpu time
0.14 seconds

CONCLUSION
Sending mail from the SAS DATA Step is a simple but powerful feature, whether you want to send a single mail with
or without an attachment or you want to send a personalized mail to all your friends or colleagues in the company.
In fact it would be easy to set up your own spamming operation. But this paper should not be seen as an
encouragement to do so. On the contrary. Spamming is illegal and using smart software like SAS to do it does not
change that.

